
CC7220-1 LA WEB DE DATOS PRIMAVERA 2023

LECTURE 7: SPARQL [1.0]

Aidan Hogan aidhog@gmail.com

LAST TIME ...



#### SEMANTIC WEB: LOGIC

#### DATA:

```
(Ireland, partOf, Europe)
(Ireland, isA, Country)
(Ireland, capital, Dublin)
```

```
Dublin

(Ireland,capital,Dublin)

(Dublin,population,1000000)
```

```
Logic: "(b, \mathsf{capital}, a) \to (a, \mathsf{partOf}, b)" "(a, \mathsf{partOf}, b), (b, \mathsf{partOf}, c) \to (a, \mathsf{partOf}, c)"
```

QUERY: "(x, partOf, y)?"

OUTPUT: 
$$\{(x \mapsto \mathsf{Ireland}, y \mapsto \mathsf{Europe}), \ (x \mapsto \mathsf{Dublin}, y \mapsto \mathsf{Ireland}), \ (x \mapsto \mathsf{Dublin}, y \mapsto \mathsf{Europe})\}$$



# TODAY'S TOPIC

## SEMANTIC WEB: QUERY

#### DATA:

```
(Ireland, partOf, Europe)
(Ireland, isA, Country)
(Ireland, capital, Dublin)
```

```
Dublin

(Ireland,capital,Dublin)

(Dublin,population,1000000)
```

```
LOGIC: "(b, \mathsf{capital}, a) \to (a, \mathsf{partOf}, b)"
"(a, \mathsf{partOf}, b), (b, \mathsf{partOf}, c) \to (a, \mathsf{partOf}, c)"
```


QUERY: "(x, partOf, y)?"

```
OUTPUT: \{(x \mapsto \mathsf{Ireland}, y \mapsto \mathsf{Europe}), (x \mapsto \mathsf{Dublin}, y \mapsto \mathsf{Ireland}), (x \mapsto \mathsf{Dublin}, y \mapsto \mathsf{Europe})\}
```






## SPARQL: QUERY LANGUAGE FOR RDF



How to ask: "Who stars in 'Sharknado'?"

## SPARQL: QUERY LANGUAGE FOR RDF



#### Query:

```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
  ex:Sharknado ex:stars ?star .
}
```

#### Solutions:

?star
ex:JohnHeard
ex:IanZiering

SPARQL: Prefix declarations

## SPARQL: PREFIX DECLARATIONS

Shortcuts for IRIs (exactly like in Turtle)

```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
  ex:Sharknado ex:stars ?star .
}
```


Specifies what to match in the data



```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
   ex:Sharknado ex:stars ?star .
}
```

"Triple pattern"

(a triple with variables)




#### Query:

```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
   ex:Sharknado ex:stars ?star .
}
```


#### Solutions:

?star
ex:JohnHeard
ex:IanZiering



How to ask: "What movies did the stars of 'Sharknado' also star in?"

### SPARQL: BASIC GRAPH PATTERNS




#### Query:

```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
  ex:Sharknado ex:stars ?star .
  ?movie ex:stars ?star .
}
```

#### Solutions:


?star	?movie
ex:IanZiering	ex:Sharknado2
ex:IanZiering	ex:Sharknado
ex:JohnHeard	ex:Sharknado

## SPARQL: BASIC GRAPH PATTERNS



### Query:


```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
    ex:Sharknado ex:stars ?star .
    ?movie ex:stars ?star .
}
```



"Basic Graph Pattern"


(a set of triple patterns)

## SPARQL: JOIN VARIABLES




#### Query:

### SPARQL: DISJUNCTION



How to ask: "What are the titles of the (first two) movies in the Sharknado series?"

## SPARQL: Disjunction (UNION)




#### Query:

```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
    { ex:SharknadoSeries ex:firstMovie ?movie . }
    UNION
    { ex:SharknadoSeries ex:secondMovie ?movie . }
    ?movie ex:title ?title .
}
```

#### Solutions:


?movie	?title
ex:Sharknado	"Sharknado"@en
ex:Sharknado2	"Sharknado 2: The Second One"@en

#### SPARQL: LEFT-JOIN



How to ask: "Give me the titles of all movies and, if available, their first-aired date?"

## SPARQL: LEFT-JOIN (OPTIONAL)



#### Query:

```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
   ?movie a ex:Movie ; ex:title ?title .
   OPTIONAL { ?movie ex:firstAired ?date }
}
```


### Solutions:

?movie	?title	?date
ex:Sharknado	"Sharknado"@en	"2013-07-11"^^xsd:date
ex:Sharknado2	"Sharknado 2: The Second One"@en	

#### "UNBOUND Variable"


(a variable without a binding in a solution)

## SPARQL: FILTERING RESULTS



How to ask: "What movies were first aired in 2014?"

### SPARQL: FILTER



#### Query:

#### Solutions:

```
?movie ?date

"Empty Results"
```

### SPARQL: FILTER

#### Query:

```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
   ?movie a ex:Movie ; ex:firstAired ?date .
   FILTER(?date > "2013-12-31"^^xsd:date
          && ?date <= "2014-12-31"^^xsd:date)
}</pre>
```

What happens in this case where ?date bound in data to a string?

FILTERs (and other functions we see later) expect certain types. If not given, a type error is given.

## SPARQL: BOOLEAN FILTER OPERATORS

- FILTERs evaluate as true, false or error
- Only results evaluating as true are returned
- Can apply AND (&&) or OR (||)
- Can also apply NOT (!)
  - $-!E \rightarrow E$

_	В	A II D	A 0 0 D
A	В	AIIB	A && B
Т	Т	Т	Т
Т	F	Т	F
F	Т	Т	F
F	F	F	F
Т	Е	Т	Е
Е	Т	Т	Е
F	Е	Е	F
Е	F	Е	F
Е	Е	Е	Е

## SPARQLOPERATORS

$\overline{A}$	Op	В	B Return type and value				
	!	BOOL $b$	BOOL $b$ BOOL true if $I_L(b)$ is false; false otherwise				
BOOL $b_1$	11	BOOL $b_2$	BOOL	true if $I_L(b_1)$ or $I_L(b_2)$ ; false otherwise			
BOOL $b_1$	&&	BOOL $b_2$	BOOL	true if $I_L(b_1)$ and $I_L(b_2)$ ; false otherwise			
TERM* $t_1$	=	TERM* $t_2$	BOOL	true if $t_1$ same term as $t_2$ ; false otherwise			
TERM* $t_1$	! =	TERM* $t_2$	BOOL	true if $t_1$ not same term as $t_2$ ; false otherwise			
$com v_1$	=	$com v_2$	BOOL	true if $I_L(v_1) = I_L(v_2)$ ; false otherwise			
$com v_1$	! =	$_{ m COM}$ $v_2$	BOOL	true if $I_L(v_1) \neq I_L(v_2)$ ; false otherwise			
$com v_1$	<	$_{ m com} v_2$	BOOL	true if $I_L(v_1) < I_L(v_2)$ ; false otherwise			
$_{ m com} v_1$	>	$_{\text{COM}} v_2$	BOOL	true if $I_L(v_1) > I_L(v_2)$ ; false otherwise			
$_{ m com} v_1$	<=	$_{ m COM}$ $v_2$	BOOL	true if $I_L(v_1) \leq I_L(v_2)$ ; false otherwise			
$_{ m com} v_1$	>=	$com v_2$	BOOL	true if $I_L(v_1) \geq I_L(v_2)$ ; false otherwise			
	+	NUM $n$	NUM	n			
	-	NUM $n$	NUM	-n			
NUM $n_1$	+	NUM $n_2$	NUM	$I_L(v_1) + I_L(v_2)$			
$_{ m NUM}$ $n_1$	_	$_{ m NUM}$ $n_2$	NUM	$I_L(v_1) + I_L(v_2)$			
NUM $n_1$	*	$_{ m NUM}$ $n_2$	NUM	$I_L(v_1) \times I_L(v_2)$			
NUM $n_1$	/	NUM $n_2$	NUM	$\frac{I_L(v_1)}{I_L(v_2)}$			

- $\bullet$  com: a comparable literal value
- $\bullet$   $_{\rm TERM}*:$  a non-comparable RDF term
- $I_L(\cdot)$ : the value (e.g., 2 not "2")

## SPARQL Functions: Existence, equality, if ...

Function	$\operatorname{Ret} olimits_{\mathfrak{p}}$	Return type and value			
$bound( exttt{TERM}\ t)$	BOOL	true if t is bound; false if unbound			
$\mathtt{if}(\mathtt{bool}\ b,\mathtt{term}\ t_1,\mathtt{term}\ t_2)$	TERM	$t_1$ if b is true; $t_2$ otherwise			
$\mathtt{coalesce}(_{\mathtt{TERM}}\;t_1,\!,\!t_n)$	TERM	TERM first $t_i$ $(1 \le i \le n)$ that is not an error or unbound			
$\mathtt{not}\ \mathtt{exists}(_{\mathtt{SUB}}\ Q)$	BOOL	bool true if $Q$ has any solution; false otherwise			
$\mathtt{exists}(_{\mathtt{SUB}}\ Q)$	BOOL	true if $Q$ has no solution; false otherwise			
$ exttt{sameTerm}( exttt{term}\ t_1,  exttt{term}\ t_2)$	BOOL	true if $t_1$ same term as $t_2$ ; false otherwise			
TERM $t$ in(TERM $t_1,, t_n$ )	BOOL	true if $t = t_i$ for any $t_i \in \{t_1,, t_n\}$ ; false otherwise			
TERM $t$ not in(TERM $t_1,,t_n$ )	BOOL	true if $t \neq t_i$ for all $t_i \in \{t_1,, t_n\}$ ; false otherwise			



• sub: a sub-query

## SPARQL FUNCTIONS: TERMS

Function	Retu	Return type and value			
isIRI(term t)	BOOL	true if t is an IRI; false otherwise			
$\mathtt{isBlank}(\mathtt{ ext{ iny TERM}}\ t)$	BOOL	true if $t$ is a blank node; false otherwise			
$\mathtt{isLiteral}(_{\mathtt{TERM}}\;t)$	BOOL	true if $t$ is a literal; false otherwise			
$ ext{isNumeric}( ext{ text{TERM}}\ t)$	BOOL	true if $t$ is a numeric value; false otherwise			
$\mathtt{str}(\operatorname{lit}\ l _{\mathtt{IRI}}\ i)$	STR	lexical value of $l \mid \text{string of } i$			
lang(LIT l)	STR	language tag string of $l$			
$\mathtt{datatype}(_{\mathtt{LIT}}\ l)$	IRI	datatype IRI of $l$			
$\mathtt{iri}(_{\mathtt{STR}}\;s _{\mathtt{IRI}}\;i)$	IRI	$s$ resolved against the in-scope base IRI $\mid i$			
$bnode([{}_{\mathtt{STR}}\ s])$	BNODE	fresh blank node [unique to $s$ ]			
$\mathtt{strdt}(_{\mathtt{STR}}\ s,_{\mathtt{IRI}}\ i)$	LIT	"s"~ <i></i>			
$\mathtt{strlang}(\mathtt{str}\ s,\!\mathtt{str}\ l)$	LIT	"s $@l$			
uuid()	IRI	fresh IRI (from UUID URN scheme)			
struuid()	STR	fresh string (from UUID URN scheme)			



- a|b indicates a or b
- [a] indicates a optional

# SPARQL Functions: Strings

Function	Retu	Return type and value			
${ t strlen}({ t str} \ s)$	INT	length of string $s$			
$\mathtt{substr}(\mathtt{str}\ s,\mathtt{int}\ b,[\mathtt{int}\ l])$	STR	substring of $s$ from index $b$ [of length $l$ ]			
$\mathtt{ucase}(\mathtt{str}\ s)$	STR	uppercase $s$			
$\mathtt{lcase}({}_{\mathtt{STR}}\;s)$	STR	lowercase $s$			
$\mathtt{strstarts}(_{\mathtt{STR}}\ s,_{\mathtt{STR}}\ p)$	BOOL	true if $s$ starts with $p$ ; false otherwise			
$\mathtt{strends}({}_{\mathtt{STR}}\; s,_{\mathtt{STR}}\; p)$	BOOL	true if $s$ ends with $p$ ; false otherwise			
$\mathtt{strbefore}({}_{\mathtt{STR}}\;s,_{\mathtt{STR}}\;p)$	STR	string before first match for $p$ in $s$			
$\mathtt{strafter}({}_{\mathtt{STR}}\ s,_{\mathtt{STR}}\ p)$	STR	string after first match for $p$ in $s$			
$\texttt{encode\_for\_iri}(\texttt{{}_{STR}}\;s)$	STR	s percent-encoded			
$\mathtt{concat}(_{\mathtt{STR}}\ s_1,, s_n)$	STR	$s_1,, s_n$ concatenated			
${ t langMatches}({ t str}\ s, { t str}\ l)$	BOOL	true if s a language tag matching $l$ ; false otherwise			
$\mathtt{regex}(\mathtt{str}\ s,\mathtt{str}\ p[,\mathtt{str}\ f])$	BOOL	true if s matches regex $p$ [with flags $f$ ]; false otherwise			
$\texttt{replace}(\texttt{str}\ s, \texttt{str}\ p, \texttt{str}\ r[, \texttt{str}\ f])$	STR	s with matches for regex $p$ [with flags $f$ ] replaced by $r$			




# SPARQL Functions: Numerics

Function	Return type and value
$\mathtt{abs}(\mathtt{NUM}\ n)$	Num absolute value of $n$
$\mathtt{round}(\mathtt{NUM}\ n)$	Num round to nearest whole number (towards $+\infty$ for *.5)
$\mathtt{ceil}(\mathtt{NUM}\ n)$	Num round up (towards $+\infty$ ) to nearest whole number
${ t floor}({ t NUM} \ n)$	Num round down (towards $-\infty$ ) to nearest whole number
rand(NUM n)	random double between 0 (inclusive) and 1 (exclusive)



## SPARQL FUNCTIONS: TEMPORAL

Function	Return type and value			
now()	рт current date-time			
$\mathtt{year}({}_{\mathtt{DT}}\ d)$	year of $d$ (as an integer)			
month(DT d)	$\underline{\text{INT}}$ month of $d$ (as an integer)			
$\mathtt{day}(\mathtt{DT} d)$	day of d (as an integer)			
hours(DT d)	hours of $d$ (as an integer)			
$minutes(\mathbf{p}\mathbf{T} d)$	$\overline{\mathbf{n}}$ minutes of $d$ (as an integer)			
$seconds(\mathbf{p}_{\mathbf{T}} d)$	$\underline{\text{INT}}$ seconds of $d$ (as an integer)			
timezone(DT d)	timezone of $d$ (as day-time-duration)			
$tz(DT\ d)$	timezone of d (as a string)			



• DT: date—time

• DTD: day—time—duration

## SPARQL Functions: Hashing

Function	Return type and value			
md5(str s)	MD5 hash of $s$			
$\mathtt{sha1}({}_{\mathtt{STR}}\;s)$	SHA1 hash of $s$			
$\mathtt{sha256}(\mathtt{str}\ s)$	SHA256 hash of $s$			
$\mathtt{sha384}(\mathtt{STR}\ s)$	SHA384 hash of $s$			
sha512(str s)	STR SHA512 hash of $s$			



### SPARQL: CASTING BETWEEN TYPES


- Y: always allowed
- N: never allowed
- M: depends on value
  - e.g., "2"^^xsd:string can be mapped to xsd:int but "P"^^xsd:string cannot

From \ To	str	flt	dbl	dec	int	dT	bool
str	Υ	М	М	М	М	M	М
flt	Υ	Υ	Υ	М	М	N	Υ
dbl	Υ	Υ	Υ	М	М	N	Υ
dec	Υ	Υ	Υ	Υ	Υ	N	Υ
int	Υ	Υ	Υ	Υ	Υ	N	Υ
dT	Υ	N	N	N	N	Υ	N
bool	Υ	Υ	Υ	Υ	Υ	N	Υ
IRI	Υ	N	N	N	N	N	N
Itri	Υ	М	М	М	М	М	М

bool = <u>xsd:boolean</u> dbl = <u>xsd:double</u> flt = <u>xsd:float</u> dec = <u>xsd:decimal</u> int = <u>xsd:integer</u> dT = <u>xsd:dateTime</u> str = <u>xsd:string</u> IRI = IRI

Itrl = simple literal

## SPARQL: WHERE CLAUSE EXAMPLE (I)




```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
    { ex:SharknadoSeries ex:firstMovie ?movie . }
    UNION
    { ex:SharknadoSeries ex:secondMovie ?movie . }
    OPTIONAL
    { ?movie ex:firstAired ?date . }
    ?movie ex:title ?title .
    FILTER(REGEX(STR(?title),"*[0-9]*"))
}
```

What solutions would this query return?

#### Solutions:

?movie	?title	?date	
ex:Sharknado2	"Sharknado 2: Th	e Second One"@en	

## SPARQL: WHERE CLAUSE EXAMPLE (II)



```
PREFIX ex: <http://ex.org/voc#>
SELECT *
WHERE {
   ?movie a ex:Movie .
   OPTIONAL
   { ?movie ex:firstAired ?date . }
   FILTER(!BOUND(?date))
}
```

What solutions would this query return?


#### Solutions:

?movie	?date
ex:Sharknado2	

Can do negation!

SPARQL: QUERY TYPES

# SPARQL: SELECT WITH \*




### Query:

# PREFIX ex: <http://ex.org/voc#> SELECT \* WHERE { ?movie a ex:Movie. ?movie ex:stars ?star . }

#### Solutions:

?movie	?star
ex:Sharknado	ex:JohnHeard
ex:Sharknado	ex:IanZiering
ex:Sharknado2	ex:IanZiering

# SPARQL: SELECT WITH PROJECTION



# Query:


PREFIX ex: <http://ex.org/voc#>
SELECT ?star
WHERE {
 ?movie a ex:Movie.
 ?movie ex:stars ?star .
}

# Solutions:

?star
ex:JohnHeard
ex:IanZiering
ex:IanZiering

(number of results returned must correspond to number of matches in data)

# SPARQL: SELECT WITH DISTINCT



Query:

PREFIX ex: <http://ex.org/voc#>
SELECT DISTINCT ?star
WHERE {
 ?movie a ex:Movie.
 ?movie ex:stars ?star .
}

Solutions: (each re

(each result row must be unique)


?star

ex:JohnHeard


ex: IanZiering

# SPARQL: SELECT WITH REDUCED

}



# SPARQL: ASK




# Query:

# PREFIX ex: <http://ex.org/voc#> ASK WHERE { ?movie a ex:Movie. ?movie ex:stars ?star . }

#### Solutions:



# SPARQL: CONSTRUCT



#### Query:


```
PREFIX ex: <http://ex.org/voc#>
CONSTRUCT { ?star ex:job ex:Actor }
WHERE {
    ?movie a ex:Movie.
    ?movie ex:stars ?star .
}
```

#### Solutions:

```
@prefix ex: <http://ex.org/voc#> .
ex:JohnHeard ex:job ex:Actor .
ex:IanZiering ex:job ex:Actor .
```

Returns an RDF graph based on the matching CONSTRUCT clause.

# SPARQL: DESCRIBE (NON-NORMATIVE FEATURE)



#### Query:

```
PREFIX ex: <http://ex.org/voc#>
DESCRIBE ?star
WHERE {
    ?movie a ex:Movie.
    ?movie ex:stars ?star .
}
```

#### Solutions:

```
@prefix ex: <http://ex.org/voc#> .
ex:JohnHeard a ex:Person .
ex:IanZiering a ex:Person .
```

Returns an RDF graph "describing" the returned results. This is an non-normative feature. What should be returned is left open.

SPARQL: SOLUTION MODIFIERS

#### SOLUTION MODIFIERS

- ORDER BY (DESC)
  - Can be used to order results.
  - By default ascending (ASC), can specify descending (DESC)
  - Can order lexicographically on multiple items
- LIMIT *n* 
  - Return only *n* results

Without ORDER BY results for queries with LIMIT or OFFSET might be non-deterministic!

- OFFSET n
  - Skip the first n results

How might we ask for the second and third most recently released movies?

```
PREFIX ex: <http://ex.org/voc#>
SELECT ?movie
WHERE { ?movie ex:firstAired ?date . }
ORDER BY DESC(?date)
LIMIT 2
OFFSET 1
```

#### SOLUTION MODIFIERS

The order of execution is always:

ORDER→ OFFSET → LIMIT

Changing the order of LIMIT/OFFSET makes no difference to the query solutions.

ORDER BY must come before LIMIT/OFFSET according to the query syntax

How might we ask for the second and third most recently released movies?

```
PREFIX ex: <http://ex.org/voc#>
SELECT ?movie
WHERE { ?movie ex:firstAired ?date . }
ORDER BY DESC(?date)
LIMIT 2
OFFSET 1
```



```
PREFIX ex: <http://ex.org/voc#>
SELECT ?movie
WHERE { ?movie ex:firstAired ?date . }
ORDER BY DESC(?date)
OFFSET 1
LIMIT 2
```

SPARQL: NAMED GRAPHS

SPARQL: NAMED GRAPHS

# SPARQL DEFINED OVER A DATASET

- A dataset  $D = (G, \{(G_1, n_1), \dots, (G_k, n_k)\})$
- $G, G_1, \ldots, G_k$  are RDF graphs
- $n_1, \ldots, n_k$  are pairwise distinct IRIs
- G is called the **default graph**
- each  $(G_i, n_i)$  is a named graph  $(1 \le i \le n)$


Core idea: SPARQL can support multiple RDF graphs, not just one.

When using SPARQL, you can partition your data into multiple graphs.

The default graph is chosen if you don't specify a graph.

Otherwise you can explicitly select a named graph using it's IRI name.

#### AN EXAMPLE DATASET



- Say we have dataset  $D = (G, \{(G_1, n_1), \dots, (G_k, n_k)\})$
- A query can pick an active dataset from the named graphs


#### FROM

- Used to define a default graph for the query using graph names
- If multiple graphs are specified, they are RDF-merged


#### FROM NAMED


Used to select the active named graphs to be used for the query

Using either feature clears the index dataset




# RDF MERGE: A QUICK REMINDER







#### How should we combine these two RDF graphs?






# RDF MERGE: A QUICK REMINDER






#### How should we combine these two RDF graphs?







- Indexed dataset:  $D = \{G, (G_1, n_1), \dots, (G_k, n_k)\}$
- Query dataset (no FROM/FROM NAMED): D
- Query dataset D' (example 1):

```
FROM n_1 FROM n_2 FROM NAMED n_3 \rightarrow D' = \{G_1 \uplus G_2, (G_3, n_3), (G_4, n_4)\} FROM NAMED n_4
```

• Query dataset D' (example 2):

$$\begin{array}{ccc} \mathsf{FROM} & n_1 \\ \mathsf{FROM} & n_2 \end{array} & \to D' = \{G_1 \uplus G_2\}$$

(⊎ indicates RDF merge)

- Indexed dataset:  $D = (G, \{(G_1, n_1), \dots, (G_k, n_k)\})$
- Query dataset (no FROM/FROM NAMED): D
- Query dataset D' (example 1):

```
FROM n_1 FROM n_2 FROM NAMED n_3 FROM NAMED n_4 \rightarrow D' = (G_1 \uplus G_2, \{(G_3, n_3), (G_4, n_4)\})
```

• Query dataset D' (example 2):

$$egin{array}{ll} {\sf FROM} \; n_1 \ {\sf FROM} \; n_2 \end{array} & o D' = \{G_1 \uplus G_2\} \end{array}$$

(\(\pmu\) indicates RDF merge)

- Indexed dataset:  $D = (G, \{(G_1, n_1), \dots, (G_k, n_k)\})$
- Query dataset (no FROM/FROM NAMED): D
- Query dataset D' (example 1):

```
FROM n_1 FROM n_2 FROM NAMED n_3 \rightarrow D' = (G_1 \uplus G_2, \{(G_3, n_3), (G_4, n_4)\}) FROM NAMED n_4
```

• Query dataset D' (example 2):


FROM 
$$n_1$$
  $\rightarrow D' = (G_1 \uplus G_2, \{\})$ 

(\(\pmu\) indicates RDF merge)

# QUERYING THE NAMED GRAPHS IN A DATASET

- We can query parts of the dataset using GRAPH
  - Specifies the IRI of a named graph over which the pattern is evaluated
  - Can also be a variable that ranges over all named graphs
  - Does not access the default graph!
  - If not specified, default graph is accessed

# AN EXAMPLE QUERY



#### Query:

```
PREFIX ex: <http://ex.org/voc#>
SELECT DISTINCT ?s
WHERE { ?s ?p ?o }
```


What solutions would this query return?

#### Solutions:

?s ex:SharknadoSeries

No GRAPH clause so answers come from default graph only

#### Using FROM



# Query:

```
PREFIX ex: <http://ex.org/voc#>
FROM ex:Sharknado.ttl
FROM ex:Sharknado2.ttl
SELECT DISTINCT ?s
WHERE { ?s ?p ?o }
```


No GRAPH clause so answers come from default graph defined by FROM (old default graph cleared)

What solutions would this query return?

#### Solutions:

?s
ex:Sharknado
ex:Sharknado2

#### Using FROM NAMED



### Query:

```
PREFIX ex: <http://ex.org/voc#>
FROM NAMED ex:Sharknado.ttl
SELECT DISTINCT ?s
WHERE { ?s ?p ?o }
```


What solutions would this query return?

#### Solutions:

?s

No GRAPH clause so answers come from default graph, which is empty (since old default graph cleared)!

#### Using GRAPH with a variable



#### Query:

```
PREFIX ex: <http://ex.org/voc#>
SELECT DISTINCT ?s ?g
WHERE { GRAPH ?g { ?s ?p ?o } }
```


What solutions would this query return?

#### Solutions:

?s	?g
ex:Sharknado	ex:Sharnado.ttl
ex:Sharknado2	ex:Sharnado2.ttl

GRAPH clause only ranges over the named graphs.

#### Using GRAPH with a name



#### Query:


```
PREFIX ex: <http://ex.org/voc#>
SELECT DISTINCT ?s
WHERE {
   GRAPH ex:Sharknado.ttl { ?s ?p ?o }
}
```

What solutions would this query return?

#### Solutions:

?s ex:Sharknado

#### Using GRAPH with FROM



#### Query:

```
PREFIX ex: <http://ex.org/voc#>
FROM ex:Sharknado.ttl
SELECT DISTINCT ?s ?g
WHERE {
   GRAPH ?g { ?s ?p ?o }
}
```


What solutions would this query return?

#### Solutions:

?s | ?g

No named graphs specified!

#### USING GRAPH WITH FROM NAMED



# Query:

```
PREFIX ex: <http://ex.org/voc#>
FROM NAMED ex:Sharknado.ttl
SELECT DISTINCT ?s ?g
WHERE {
   GRAPH ?g { ?s ?p ?o }
}
```


What solutions would this query return?

#### Solutions:

?s	?g
ex:Sharknado	ex:Sharnado.ttl

**GRAPH** accesses the one and only named graph

#### USING GRAPH WITH FROM AND FROM NAMED



#### Query:

```
PREFIX ex: <http://ex.org/voc#>
FROM ex:Sharknado2.ttl
FROM NAMED ex:Sharknado.ttl
SELECT DISTINCT ?x ?q
WHERE {
   GRAPH ?g { ?s ?p ?o }
   ?x ?q ?o .
}
```

What solutions would this query return?

#### Solutions:

?x	?q
ex:Sharknado2	ex:stars



